Question: I have a big sorted list and some of it is important, I want to visualize the position of these important elements.

```
RandomSeed[1234];
red = RandomInteger[10000, 100];(*important elements*)
point = Sort@RandomInteger[10000, 1000];(*all elements*)
pointColor = If[MemberQ[red, #], Style[#, Red], #] & /@ point;
ListPlot[pointColor]

```

[![enter image description here](https://i.stack.imgur.com/yqfaR.png)](https://i.stack.imgur.com/yqfaR.png)

I try some ways fails such as 

* `ListPlot[pointColor, PlotStyle -> RGBColor[0, 0, 1, 0.04]]`
* `pointColor = If[MemberQ[red, #], Style[#, Red, Thick], #] & /@ point;
ListPlot[pointColor]`

but only this code:

```
ListPlot[pointColor, PlotStyle -> Blue] /. 
 RGBColor[0, 0, 1] -> RGBColor[0, 0, 1, 0.04]

```

[![enter image description here](https://i.stack.imgur.com/ZlZHN.png)](https://i.stack.imgur.com/ZlZHN.png)

So what the better way to visualize these important elements?

Answer: 
```
pointColor = If[MemberQ[red, #], Style[#, Opacity[1, Red], PointSize[Medium]], #]& /@ point;

ListPlot[pointColor, PlotStyle -> RGBColor[0, 0, 1, 0.04]]

```

[![enter image description here](https://i.stack.imgur.com/2ARAT.png)](https://i.stack.imgur.com/2ARAT.png)
```
RandomSeed[1234];
red = AssociationThread[RandomInteger[10000, 100] -> True];
point = Sort@RandomInteger[10000, 1000];

```

Get the positions `pos1` of important points:

```
pos1 = Pick[Range[Length[point]], Lookup[red, point, False]];
pos2 = Complement[Range[Length[point]], pos1];

```

Put the important points last in `Show` to get them on top of the other points:

```
Show[
 ListPlot[Thread[{pos2, point[[pos2]]}], PlotStyle -> Directive[Black, PointSize[Small]]],
 ListPlot[Thread[{pos1, point[[pos1]]}], PlotStyle -> Directive[Red, PointSize[Large]]]]

```

![](https://i.stack.imgur.com/Vz8U4.png)